

HP-41C 82162A Printer Barcode by Winfried Maschke  
PPC Calculator Journal V9 No 4 Pg 45 May-Jul 1982  
HP-41C & CV need Extended Functions/Memory Module

```
01 LBL "BAP" ;HP82162A Barcode Printing Program
02 CF 10 ;Flag 10 OFF = Non-Private Barcode
03 GTO 04 ;Start Printing (Line 6)
04 LBL "BAPV" ;Private Barcode
05 SF 10 ;Flag 10 ON = Private Barcode
06 LBL 04 ;Start Printing
07 CF 12 ;Single Width
08 XROM 29,01 ;ACA
09 ADV ;Print Title
10 CLRG ;Clear Registers
11 ABS ;ABS (Bytes Per Row Entered in X)
12 INT ;INT (Bytes Per Row)
13 X=0?
14 SIGN ;Minimum Bytes/Row = 1
15 13
16 X>Y?
17 X<>Y ;Maximum Bytes/Row = 13
18 STO 06 ;Bytes Per Row (1-13)
19 12
20 + ;Data Registers = Bytes Per Row + 12
21 XROM 25,30 ;PSIZE - Set Data Register Size
22 E
23 - ;Subtract 1
24 E3
25 / ;Divide by 1000
26 STO 03 ;Register Pointer = 0.0XX (XX=12-24)
27 SF 25 ;Error Ignore Flag ON
28 XROM 25,38 ;SAVEP - Copy Program to X-Memory
29 FIX 0 ;Integer only
30 CF 29 ;Digit Grouping Flag OFF
31 XROM 25,34 ;RCLPTA - Recall Program Length
32 2
33 -
34 STO 05 ;Number of Bytes - 2 (w/o Checksum)
35 LASTX ;2
36 + ;Number of Bytes Including Checksum
37 XROM 29,05 ;ACX - Accumulate X into Print Buffer
38 XROM 25,15 ;FLSIZE - Number of 7-Byte Registers in File
39 X<>Y
40 " BYTE"
41 XROM 29,01 ;ACA - Accumulate Alpha into Print Buffer
42 ADV ;Print XX BYTE
43 RCL 06 ;Bytes Per Row (1-13)
44 / ;Byte Count / Bytes Per Row
45 INT ;INT (Byte Count / Bytes Per Row)
46 LASTX ;Byte Count / Bytes Per Row
47 FRC
48 X#0? ;If Remainder (Byte Count / Bytes Per Row)
49 SIGN ;Round Up to the Next Row
50 +
51 "="
52 XROM 29,01 ;ACA
53 XROM 29,05 ;ACX
54 " ROWS"
55 XROM 29,01 ;ACA
56 ADV ;Print = XX ROWS
57 ADV ;Line Feed
58 SF 12 ;Double Width
59 RDN ;Number of 7-Byte Registers in File
60 "\20\BE\00\00\00"
61 256
62 / ;RH = High Byte of 7-Byte Register Count
63 XROM 25,47 ;XTOA - Convert X and Append to Alpha
64 LASTX ;256
65 * ;Count of 7-Byte Registers in File
66 LASTX ;256
```

```

67 MOD          ;RL = Low Byte of 7-Byte Register Count
68 XROM 25,47   ;XTOA - Convert X and Append to Alpha
69 RCL M        ;Recall 20BE000000RHRL to X as NNN
70 XEQ 06       ;Curtain Moving Routine (Line 86)
71 STO 06       ;Store BE000000RHRL into Register 06
72 SF 08
73 E
74 STO 01       ;Register 01 set to 1
75 XEQ 08       ;Reset Register Pointer (Line 101)
76 "\C0\00\2F" ;Permanent END
77 3
78 STO 04       ;Register 04 set to 3
79 STO 05       ;Register 05 set to 3
80 XEQ 11       ;Process Bytes (Line 124)
81 FS?C 09
82 XEQ 05       ;Create Barcode Row Header (Line 152)
83 RCL 06       ;Bytes Per Row (1-13)
84 BEEP
85 XROM 28,36   ;PWRDN - Set All Devices to Low Power
86 LBL 06       ;Curtain Moving Routine
87 "\0B\E0\01\69\0B\E0\B0"
88 RCL M        ;0BE001690BE0B0 (NNN in X)
89 X<> c        ;Sigma=0BE, R00=0BE, .END.=0B0
90 X<>Y        ;X=20BE000000RHRL, Y=Old contents of c
91 X<> 00       ;X=Old contents of R00, R00=20BE000000RHRL
92 X<>Y        ;X=Old contents of c, Y=Old contents of R00
93 X<> c        ;X=0BE001690BE0B0, c=Old contents restored
94 X<>Y        ;X=Old contents of R00, Y=0BE001690BE0B0
95 RTN
96 GTO 04       ;Start Printing (Line 6)
97 LBL 15       ;Get Next Register
98 ISG 03       ;Increment Register Pointer
99 GTO 02       ;Decrement Byte Count (Line 113)
100 XEQ 05      ;Create Barcode Row Header (Line 152)
101 LBL 08       ;Reset Register Pointer
102 RCL 03       ;Register Pointer 12. to 24.0XX (XX=12 to 24)
103 FRC
104 12
105 +
106 STO 03       ;Reset Register Pointer = 12.0XX (XX=12 to 24)
107 16           ;Multiplier to Shift Bits Left by 4
108 RCL 01       ;Instruction Length Counter
109 E           ;1
110 -           ;Decrement Instruction Length
111 *           ;Shift to Upper 4 Bits - Leading Broken Fn Bytes
112 STO 11       ;Number of Leading Broken Function Bytes
113 LBL 02       ;Decrement Number of Bytes Remaining in Program
114 DSE 05       ;Bytes Remaining = Bytes Remaining - 1
115 GTO 02       ;Process Next Byte Group (Line 117)
116 RTN
117 LBL 02       ;Process Next Byte Group from File
118 DSE 04       ;0 to 6 Bytes Remaining
119 GTO 11       ;If 1 to 6 Bytes, Process Bytes
120 7           ;Get Next 7-Byte Register from File
121 STO 04       ;Register 04 set to 7
122 XROM 25,23   ;GETX - Copy File Register to X
123 STO M        ;Synthetic Write to Alpha Register
124 LBL 11       ;Process Bytes
125 SF 09
126 XROM 25,01   ;ALENG - Return Alpha String Length
127 RCL 04       ;Number of Bytes Remaining (1 to 7)
128 X#Y?        ;String Length Not Equal To Bytes Remaining
129 .           ;0
130 X=Y?        ;Must Fail if Previous Test (X#Y?) is True
131 XROM 25,07   ;ATOX - Convert Alpha Character to X
132 ENTER
133 STO IND 03   ;Store Instruction Byte in Next Slot (12-24)
134 ST+ 09       ;Add to Checksum
135 DSE 01       ;Decrement Instruction Length Counter
136 GTO 15       ;Get Next Register (Line 97)

```

```

137 143      ;Instructions 00-8F (000-143) = 1 Byte
138 -
139 X<=0?
140 97
141 64      ;Instructions 90-CF (144-207) = 2 Bytes
142 -
143 X<=0?
144 34
145 32      ;Instructions D0-EF (208-239) = 3 Bytes
146 -
147 X<=0?      ;Instructions F0-FF (240-255) = 1-16 Bytes
148 3
149 STO 01      ;Store Instruction Length in Register 01
150 STO 07      ;Store Instruction Length in Register 07
151 GTO 15      ;Get Next Register (Line 97)
152 LBL 05      ;Create Barcode Row Header
153 CF 09
154 RCL 08      ;Row Sequence Number
155 16
156 MOD      ;Row Sequence MOD 16 for Barcode Header
157 LASTX      ;16
158 FS? 10      ;Private Barcode?
159 ST+ X      ;X+32 = Type 2 = Private
160 +      ;X+16 = Type 2 = Non-Private
161 ST+ 09      ;Add to Checksum
162 STO 10      ;2nd Byte of BC Header = Type & Seq. #
163 ISG 08      ;Increment Row Sequence Number
164 ""      ;F0=null string (NOP)
165 RCL 07      ;Instruction Length
166 RCL 01      ;Instruction Length Counter
167 DSE X      ;Decrement Instruction Length
168 -      ;Number of Broken Trailing Function Bytes
169 RCL 11      ;Number of Broken Leading Function Bytes
170 +      ;Add to Broken Leading to Trailing Function Bytes
171 STO 11      ;Number of Broken Leading/Trailing Function Bytes
172 RCL 09      ;Running Checksum
173 +      ;Add Byte Value to Checksum
174 255
175 MOD      ;Checksum MOD 255
176 X=0?      ;If Checksum is Zero
177 LASTX      ;Reset to 255
178 STO 09      ;Save Running Checksum in Register 9
179 ASTO 02      ;Store Remainder of Instruction String in 02
180 RCL 03      ;Register Pointer 13. to 25.0XX (XX=12 to 24)
181 INT      ;13 to 25 (Value After ISG 03 is True)
182 8999      ;Add 8999
183 +      ;9012 to 9024
184 E3      ;1000
185 /      ;9.012 to 9.024
186 STO 00      ;Register Pointer 9 to 12 through 24
187 LBL 12      ;Print Barcode Row
188 RCL 08      ;Row Sequence Number
189 XROM 29,05 ;ACX
190 ADV
191 ADV
192 XROM 28,36 ;PWRDN - Set All Devices to Low Power
193 XROM 28,37 ;PWRUP - Set All Devices to Operating Power
194 XROM 28,27 ;AUTOIO - Set Interface to Auto Mode
195 SF 17      ;Record Incomplete Flag
196 CLX
197 XROM 29,03 ;ACCOL
198 118
199 RCL 03      ;Register Pointer 13. to 25.0XX (XX=12 to 24)
200 INT      ;13 to 25 (Value After ISG 03 is True)
201 +      ;131 to 143 (Code Sent to Printer after ESC 1)
202 "\1B\31" ;ESC 1 (Sets Printer for Barcode Printing)
203 XROM 25,47 ;XTOA - Convert X and Append to Alpha
204 XROM 28,35 ;OUTA - Output Alpha to Primary Device
205 RCL 00      ;Register Pointer 9.012 to 9.024
206 LBL 13      ;Print a Byte of Barcode

```

```

207 RCL IND X ;Recall Bytes in Registers 9 to 12 through 24
208 CLA
209 XROM 25,47 ;XTOA - Convert X and Append to Alpha
210 SF 25 ;Error Ignore Flag
211 X=0?
212 XROM 29,03 ;ACCOL
213 X#0?
214 XROM 28,35 ;OUTA - Output Alpha to Primary Device
215 RDN
216 XROM 28,31 ;INSTAT - Input Status from Primary Device
217 FS? 03 ;Paper Out?
218 GTO 14 ;Go to Paper Out Message
219 RDN
220 ISG X ;Increment Register Pointer
221 GTO 13 ;Print a Byte of Barcode (Line 206)
222 ADV ;Put Space Between Barcode Lines
223 ADV
224 ADV
225 ADV
226 ADV
227 ADV
228 CLA
229 ARCL 02 ;Recall Remainder of Instruction String
230 RTN
231 LBL 14 ;Paper Out Message
232 XROM 28,36 ;PWRDN - Set All Devices to Low Power
233 "PAPER"
234 BEEP
235 BEEP
236 PROMPT ;Request a New Roll of Paper
237 RCL 00 ;Register Pointer 9 to 12 through 24
238 GTO 12 ;Print Barcode Row (Line 187)
239 END

```

```

0192C600F400424150A90AB500C801F50042415056A80A05A9
0CA7418F8A6168677A1113457136111240A65E1B411B134333
A819A6669C00A91DA6621241357640A745A64F71F520425954
45A7418F2643687669637A40F13DA741A745F520524F5753A7
418F8FA80C75F520BE00000012151643A66F7642764BA66F90
75E0000636A8081B31E00008F3C0002F133435E0000BAA09E0
00052686A72407F70BE001690BE0B09075CE7D71CE0071CE7D
7185B500CF0F9603B300E00005092369111240331116211B41
423B039705B30085039704BC001734A65791750CA809A64124
791A78A64783918392099701D0008F111413417B1917001614
417B1314001312417B133137D0008F06A9092811164B76AC0A
92734092093A9608F027219773412B403B29401215154B6776
399A02236818191919401B1343300D28A7458F8FA724A725A7
1BA81177A743111118236840F21B31A66FA723200E90F387A6
6FA81967A74363A72375A71FAC03BF00759673BE008F8F8F8F
8F8F879B02850FA724F5504150455286868E20BD0000000000
00C6372DC8

```

402 BYTES

# HP-41C 82162A Barcode by Winfried Maschke PPC V9 N4 P45 May-Jul 1982

Program Registers Needed: 58


Row 1 (1 - 4)



Row 2 (4 - 8)



Row 3 (9 - 19)




Row 4 (20 - 28)



Row 5 (29 - 37)



Row 6 (38 - 43)



Row 7 (44 - 53)




Row 8 (54 - 59)



Row 9 (60 - 64)



Row 10 (65 - 72)



Row 11 (73 - 80)



Row 12 (80 - 87)



Row 13 (87 - 91)



Row 14 (91 - 99)



Row 15 (99 - 107)



Row 16 (108 - 118)



# HP-41C 82162A Barcode by Winfried Maschke PPC V9 N4 P45 May-Jul 1982

Row 17 (118 - 126)



Row 18 (126 - 135)




Row 19 (135 - 141)



Row 20 (141 - 150)



Row 21 (151 - 158)



Row 22 (159 - 167)



Row 23 (168 - 178)



Row 24 (179 - 186)




Row 25 (187 - 195)



Row 26 (195 - 202)



Row 27 (203 - 210)



Row 28 (211 - 218)



Row 29 (219 - 229)



Row 30 (229 - 235)



Row 31 (236 - 239)

